



MILATARI NEWSLETTER

Volume 2 Number 10


September 1983

Price \$1.00

** NEXT MEETING **

SATURDAY, September 17th - 2PM

ARMBRUSTER SCHOOL - GREENDALE



ATARI® 800XL

A powerful, versatile tool for work and play, the ATARI 800XL Home Computer features built-in ATARI BASIC Programming Language, a full-stroke keyboard, a HELP key and an international character set—plus built-in 64K RAM memory. This additional memory gives users access to over 2,000 software programs, while an expansion connection provides the potential for adding sophisticated peripherals. Teamed with the right ATARI programs and peripherals, the ATARI 800XL Home Computer can teach, entertain, program, and help manage a home or office.

- o MEMORY: 64K RAM
 - 24K ROM (operating system plus ATARI BASIC programming language)
- o KEYBOARD: Full-stroke design. 62 keys, including HELP key and 4 special function keys. International character set. 29 graphics keys.
- o CPU: 6502C microprocessor. Clock speed of 1.79 MHz
- o SPECIAL ATARI INTEGRATED CIRCUITS: GTIA (graphic display). POKEY (sound generator and controller ports). ANTIC (controls screen and input/output).
- o PROGRAMMING FEATURES: Built-in ATARI BASIC programming language. HELP key (provides additional information and menu screens). Software compatibility (works with programs designed for all ATARI Home Computers).
- o DISPLAY: 11 graphics modes. 256 colors (128 colors displayable at one time). Maximum 320 x 192 resolution in graphics modes. 5 text modes. Maximum text display is 40 columns x 24 lines.
- o SOUND: 4 independent sound voices. 3 1/2 octave range.
- o INPUT/OUTPUT: Software cartridge slot. Expansion connection (external processor bus for memory expansion and adding future peripherals). TV output. Monitor output. 2 controller ports. Serial I/O connector.

(Features and specifications subject to change)

MILATARI * * * SEPTEMBER 1983

Milwaukee Area ATARI Users Group

This newsletter is written and printed by members of the Milwaukee Area ATARI Users Group (MILATARI), an association of individuals with a common interest in using and programming ATARI computers. MILATARI is not affiliated with the ATARI company, nor any other commercial organizations.

All articles are written and donated by the membership. Opinions expressed in this publication are those of the individual author and do not necessarily represent, nor reflect, the opinions of MILATARI nor those of any other commercial or non-commercial organizations. Any article appearing in this newsletter may be reproduced, providing credit is given to the author and to MILATARI.

Write MILATARI Newsletter at P.O. Box 1191, Waukesha, WI 53187.

MEMBERSHIP INFORMATION

Membership is open to individuals and families who are interested in using and programming ATARI computers. The membership includes the subscription to this newsletter and access to the user's library. The membership fee is \$15 per year for individual, \$20 for family and \$10 for associate. Contact Larry Leskovsek, Treas. at 547-0249 or write MILATARI, P.O. Box 1191, Waukesha, WI 53187 for more information.

MEETING INFORMATION

MILATARI meetings are held once monthly. This month the meeting will be held at the Armbruster School, 7000 Greenway, Greendale, WI. The meeting is held in the multi-purpose room. BASIC classes begin at 2:00 P.M. Technical sessions are also held at 2:00 P.M. The business session begins at 3:00 P.M. followed by demonstrations. The library will be open before and after the business meeting.

MILATARI Officers:

President	Gary Nolan 353-9716
Vice-president	Chris Stieber 529-2663
Treasurer	Larry Leskovsek 547-0249
Secretary	Jim Comaris 353-3447
Education	Linda Scott 466-2314
Chairperson	Ron Friedel 354-1717
Cassette	Dennis J. Bogie 968-9341
Librarian	Sharon Samache 421-2887
Membership	Steve Booth 367-8739
Committee	Karl Buschhaus 774-2576
Disk	David Frazer 542-7242
Librarian	Pete Kurth
Publications	
Librarian	
Newsletter	
Editor	
Bulletin Board	
SYSOP	

Technical support Group:

The following members have indicated a willingness to assist MILATARI members.

William Lawrence	1-968-3082 Programming
Don Wilcox	228-1650 Programming
Erik Hanson	252-3146 Prog/Tech
Gary Nolan	353-9716 Prog/Tech
Steve Booth	367-8739 Programming
Nick Liberski	782-5594 Prog/Tech

MILATARI Bullentin Board:

The MILATARI Users Group maintains a 24 hr bulletin board service. The phone number is 353-6031. (Beginning September 19th!)

President's Ram

by Gary Nolan

RINNNNG-RINNNNG

Hear the bells? Thems school bells! And at the September meeting we'll be starting our Basic classes, once again. If you're interested in learning to program your computer be there bright and early (about 2pm) on the 17th, and Linda Scott and friends will teach you all you ever wanted to know about Atari basic. If you're not attending the programming classes you might want to stop in at the workshop. This months workshop will cover the individual pieces of hardware that make up a total system. Also covered will be some of the more interesting additions to the hardware. We will continue to run the KIDS KORNER as long as we get the kind of response we recieved last month. It proved to be very popular. We just may have an ATARI 1050 disk drive to show, if it hasn't been sold. At the very least I'll have a review for you as I'm supposed to get my tiny fingers on one this week. Seeing as today is Labor day chances are it'll be gone right quick. (if all the bugs were fixed)

UPGRADING

Before the next meeting I should know the proper procedure for upgrading the ATARI Basic cartridge to Basic2 (or whatever they'll call it). Right now ATARI says that you'll have to send the cart to one of thier repair centers. Knowing the Post Office, and the backlog that ATARI will run into, this could take at least a month. If it's a simple switch of chips you might be able to do it yourself or have a local repair center, like Mr. TV or Kenwood TV do it. Or the club could arrange an upgrading session for members with chips. The cost is supposed to be \$15. Don't start packing them up until I get more info from ATARI, on or about the 15th of the month.

IT WORKS!!

For you people who have Letter Perfect and either a Prowriter or NEC8023 and are having trouble getting the underline function to work using the Control U option, LISTEN UP. Re-configure the printer driver disk leaving the Backspace value at ZERO not 08. If you want to b/s the printer use the Control V option. And if you didn't already know it, you can put this driver on your files disk as LJK does not write anything to sector one (where the diver is stored) and only reads it when looking for the driver.

Also for those of you with the NEC printers that have that annoying whine to them, there is a fix. You'll have to change a capicitor on the circuit board. And if memory serves me only two or three people ran into this problem. So if they will call me I'll send them a copy of the article that appeared in the Lompoc/Santa Maria UG newsletter.

THE REAL DIFFERENCE BETWEEN MEN AND BOYS IS NOT THIER AGE BUT THE PRICE OF THIER TOYS

For \$400 you can add a 128K Ramdisk unit with printer buffer/spooler and battery back-up capability to your 400/800. This unit hooks up to the computer via the joystick ports eliminating the 850. It transfers data at approx. 8K/sec. It's software controlled and a 512K upgrade will be available soon. Available from:

PROTRONICS (213) 709-2820

21115 Devonshire St.

Chatsworth, CA 91311

Yet another drive manufacturer announces a drive for the ATARI. This one is from TRAK Microcomputer Corp. and is model AT-D2. It is double density and has a printer port built in, track read-out, write protect switch and read-write and density indicators. With thier optional TURBO CHARGED software transfer speeds up to 18 times faster than any other ATARI drive is possible. List price is under \$500, discounted around \$380. For more info you could call 1-800-323-4853.

And from the Penney's Christmas catalog comes a 13" monitor/TV with remote control for \$400 or a 19" monitor/TV with remote control and a 5 watt/chan. stereo amp for \$500. They have the 600XL listed at \$150 and the 800XL at \$300, but don't try to order either computer yet.

HELP!!

A Mark Kutnyak is looking for help with a Blackjack program he's writting. Someone familiar with math routines to figure percentages based on the cards left is needed. Give Mark a call after 4:30 at 483-2313.

For you really good programmers here's a chance to make a couple of bucks. Chuck Bleau is looking for someone to write a program that will turn on some Air Raid sierens and check them to make sure they did in fact sound the warning. It should also give some type of graphic representation of the results. This will be a used in a commercial system nation wide, so you better know your stuff if you want to tackle this one! Contact Chuck at 242-2800 Ext 22 during the day and if you

MILATARI * * * SEPTEMBER 1983

can't call during the day call me for his home phone number.

MILATARI will be taking part in the HAMFEST at the WAUKASHA EXPO CENTER on Oct. 9th and needs volunteers to help man the booth. It's a one day show so it could be divided into one or two hour shifts with no problem. See me at the meeting or call if you can't make it on the 17th but want to help.

We also need people to help with the workshops. We are going to run a word processing and a printer w/s, and would like to have somebody or bodys to give a short (5/10 min.) presentation on the following;

Letter Perfect	Gemini 10/10x
Atariwriter	Epson (any)
Textwizard	Atari 825/1025/1027
Bankstreet Writer	Okidata
NEC/Prowriter	Others ?

or any printer or WP that you have. This will not be in front of the whole group but for the small workshop sessions. So those of you who are nervous in front of large groups relax! You'll do just fine. Nobody expects a well versed, articulate, captivating speech like mine (WHAT!) from everyone. So give it a try, you might like it enough to do it again.

GOOD-BYE, THANKS--HELLO, GOOD LUCK

Bill Simotti the BBS Sysop is turning over those duties to Pete Kurth this month. The old BBS will go down the 9th and the new one will be up and running on the 15th. The new phone number is 355-6031. Do NOT try to call until the 15th as the line will not be installed until then.

We'd like to thank Bill for all the time and effort he put into the BBS as well as using his equipment all this time. We realize it put a strain on both, and more work was involved than most realized, so THANKS BILL!!!!

The new system will be run on a club owned computer but we will be using Pete's DSDD Percom drives. That is unless somebody would like to donate some equipment to the group. We hope to keep it up 24hrs but that's not always possible. So you people with modems lets use and support this board, it can only grow and become useful with your help.

FOR SALE

We should have more disks for sale at the next meeting. Same brand but double density at \$19 a box of ten. Who there might even be some cassettes there too.

Prices for disks from the PD library will now be \$4 per disk. Steve should have the backlog taken care of and have a good stock of the more popular ones on hand.

BYE

Well, see you on the 17th. Bye.....

September Meeting Adgenda

1:00 PM	Officers meeting	Room 109
2:00 PM	BASIC CLASS	Room 109
	First session of a new cycle	
2:00 PM	WORKSHOP - 2nd session	ROOM 107
	more on hardware	
2:00 PM -	KIDS KORNER open	ROOM ???
5:30 PM		
2:30 PM	Technical Session	Gym
3:30 PM	BUSINESS MEETING	Gym
4:45 PM	DEMONSTRATIONS	
	Business	Gym
	Education	Room 107
	Games	Kids Korner

Cassette, Disk and Publication libraries will be open before and after business meeting.

ADVANCED GRAPHICS

Examples and discussions of Player-Missile Graphics
and features of Graphics Mode 8

- 1) Moving a Player
- 2) Setting Priority
- 3) Collision Detection
- Items 4 thru 8 will appear in the next issue.
- 4) Using Missiles
- 5) String-Player
- 6) Color Artifacts
- 7) Text in mode 8
- 8) VBLANK Player Move

Information provided by:

ATARI INC.
CONSUMER PRODUCT SERVICE
PRODUCT SUPPORT GROUP

DEMOPAC #5

PLAYER-MISSILE GRAPHICS
Moving a Player
JB 8/82

The programs on the following pages, Moving a Player, versions 1 and 2, illustrate the use of the joystick in moving a player around the screen. Both programs are entirely in BASIC.

In these, and in all of the following P/M Graphics programs, a subroutine is used to initialize the player. The subroutine performs the major housekeeping tasks for setting up a player:

- 1) P/M Graphics is enabled at DMACTL and GRACCTL, selecting single-line resolution;
- 2) An area of free RAM is allocated for player data by setting PMBASE;
- 3) The starting location of the player is computed from PMBASE;
- 4) The color and horizontal position are initialized;
- 5) The player data area is cleared.

In these programs, the RAM area selected for the players is computed from RAMTOP, the top of memory pointer. Since RAMTOP is also used in setting up the display list and screen data area, the program steps back a number of pages from RAMTOP in order to place the player data area directly below screen data area. This is not the only way to do it; you can simply select an area of free RAM. For example, to start the player data area at page 14 of memory (\$1400, or decimal 5120), simply POKE PMBASE,14*1024/256.

The body of the program reads the joystick and moves the player image. The image is not drawn on the screen except as part of the vertical movement routine, so it does not appear until the stick indicates a move down. An ON...GOSUB statement is used to read the stick, eliminating the need for time-consuming IF statements.

In version 1 of the program, a single byte of data (231) is used for all 8 lines of the player. This simplifies the vertical movement, and speeds it up considerably. Because all 8 lines have the same bit pattern, it is only necessary to erase the top line and draw the bottom line to move down, or vice-versa to move up. Version 2 illustrates the more realistic situation, in which different lines have different bit patterns, creating an interesting shape. The data for the player shape is contained in line 120. It takes noticeably longer to redraw the entire player for each vertical movement.

Using BASIC to move players vertically is quite slow. If more than one player is used, movement is even slower. Horizontal movement is much faster, since it is a simple matter of poking a register. If fast vertical movement is required, it is best to use a machine-language subroutine. An example of a VBLANK routine to read the joystick and move a player is provided on a later page.

```

1 REM MOVING A PLAYER
2 REM JB 5/82
3 REM -- select one of the four players and move it around the screen
4 REM using the joystick
5 REM *****
10 GOSUB 1000:REM initialize player
20 UP=100:DOWN=200:EAST=300:WEST=400
30 SE=500:NE=550:SW=600:NW=650:REM motion routine locations
40 Z=90:REM no motion, return only

```


MILITARI * * * SEPTEMBER 1983

```

50 REM the following statement checks the joystick, and send control
55 REM to the appropriate subroutine.
60 ON STICK(0) GOSUB Z,Z,Z,Z,SE,NE,EAST,Z,SW,NW,WEST,Z,DOWN,UP,Z
70 IF STRIG(0)=0 THEN GOSUB 800:REM on trigger, change the player color
80 GOTO 60:REM keep checking stick
90 RETURN:REM no motion, keep checking stick
95 REM *****
99 REM -- motion routines --
100 X=X-1:IF X<0 THEN X=0:REM *** move up
110 POKE PSTART+X,231:POKE PSTART+X+8,0:RETURN
111 REM draw top line, erase bottom line
200 X=X+1:IF X>250 THEN X=250:REM *** move down
210 POKE PSTART+X,231:POKE PSTART+X-8,0:RETURN
211 REM draw bottom line, erase top line
300 H=H+1:IF H>200 THEN H=200:REM *** move east
310 POKE HPOS,H:RETURN
400 H=H-1:IF H<50 THEN H=50:REM *** move west
410 POKE HPOS,H:RETURN
500 GOSUB DOWN:GOSUB EAST:RETURN:REM *** move southeast
550 GOSUB UP:GOSUB EAST:RETURN:REM *** move northeast
600 GOSUB DOWN:GOSUB WEST:RETURN:REM *** move southwest
650 GOSUB UP:GOSUB WEST:RETURN:REM *** move northwest
700 REM *****
799 REM this subroutine changes the player color
800 C=PEEK(COL):C=C+2
810 IF C>255 THEN C=4:REM skip black
820 POKE COL,C
830 RETURN
990 REM *****
999 REM this subroutine initializes the player
1000 GRAPHICS 5:PRINT "PLAYER 1,2,3,4...":INPUT PNUM:REM select a player
1005 REM assign start address, color register, horizontal position reg
1010 IF PNUM=1 THEN OFFSET=1024:COL=704:HPOS=53248
1020 IF PNUM=2 THEN OFFSET=1280:COL=705:HPOS=53249
1030 IF PNUM=3 THEN OFFSET=1536:COL=706:HPOS=53250
1040 IF PNUM=4 THEN OFFSET=1792:COL=707:HPOS=53251
1050 POKE 559,62:POKE 53277,3:REM enable players w/single-line resolution
1060 PMBASE=PEEK(106)-24:POKE 54279,PMBASE:REM step back 24 pages from
1061 REM ramtop to assign player ram area
1070 PSTART=PMBASE*256+OFFSET:REM starting address of player
1080 POKE COL,88:H=50:POKE HPOS,H:REM assign color, horizontal position
1090 FOR I=0 TO 255:POKE PSTART+I,0:NEXT I:REM clear player
1095 PRINT "MOVE STICK DOWN TO MAKE PLAYER APPEAR"
1100 RETURN

1 REM MOVING A PLAYER: VERSION 2
2 REM JB 5/82
3 REM -- select a player and move it with joystick: in this version the
4 REM player is assymetric, so the movement is slower. 7 lines are drawn
5 REM each time instead of 1 line
6 REM *****
10 GOSUB 1000:REM initialize player
20 UP=100:DOWN=200:EAST=300:WEST=400
30 SE=500:NE=550:SW=600:NW=650:REM motion routine locations
40 Z=90:REM no motion, return only
50 REM the following statement checks the joystick, and send control
55 REM to the appropriate subroutine.
60 ON STICK(0) GOSUB Z,Z,Z,Z,SE,NE,EAST,Z,SW,NW,WEST,Z,DOWN,UP,Z
70 IF STRIG(0)=0 THEN GOSUB 800:REM on trigger, change the player color
80 GOTO 60:REM keep checking stick
90 RETURN:REM no motion, keep checking stick
95 REM *****
99 REM -- motion routines --
100 X=X-1:IF X<0 THEN X=0:REM *** move up
110 FOR I=0 TO 6:READ B:POKE PSTART+X+I,B:NEXT I
111 POKE PSTART+X+7,0:RESTORE 120:RETURN
120 DATA 126,231,195,219,195,231,126
200 X=X+1:IF X>250 THEN X=250:REM *** move down
210 FOR I=0 TO 6:READ B:POKE PSTART+X+I,B:NEXT I
211 POKE PSTART+X-1,0:RESTORE 120:RETURN
300 H=H+1:IF H>200 THEN H=200:REM *** move east
310 POKE HPOS,H:RETURN
400 H=H-1:IF H<50 THEN H=50:REM *** move west
410 POKE HPOS,H:RETURN
500 GOSUB DOWN:GOSUB EAST:RETURN:REM *** move southeast
550 GOSUB UP:GOSUB EAST:RETURN:REM *** move northeast

```



```

600 GOSUB DOWN:GOSUB WEST:RETURN:REM *** move southeast
650 GOSUB UP:GOSUB WEST:RETURN:REM *** move northwest
700 REM *****
799 REM this subroutine changes the player color
800 C=PEEK(COL):C=C+2
810 IF C>255 THEN C=4:REM skip black
820 POKE COL,C
830 RETURN
990 REM *****
999 REM this subroutine initializes the player
1000 GRAPHICS 5:PRINT "PLAYER 1,2,3,4...":INPUT PNUM:REM select a player
1005 REM assign start address, color register, horizontal position reg
1010 IF PNUM=1 THEN OFFSET=1024:COL=704:HPOS=53248
1020 IF PNUM=2 THEN OFFSET=1280:COL=705:HPOS=53249
1030 IF PNUM=3 THEN OFFSET=1536:COL=706:HPOS=53250
1040 IF PNUM=4 THEN OFFSET=1792:COL=707:HPOS=53251
1050 POKE 559,62:POKE 53277,3:REM enable players w/single-line resolution
1060 PMBASE=PEEK(106)-24:POKE 54279,PMBASE:REM step back 24 pages from
1061 REM ramtop to assign player ram area
1070 PSTART=PMBASE*256+OFFSET:REM starting address of player
1080 POKE COL,88:H=50:POKE HPOS,H:REM assign color, horizontal position
1090 FOR I=0 TO 255:POKE PSTART+I,0:NEXT I:REM clear player
1095 PRINT "MOVE STICK DOWN TO MAKE PLAYER APPEAR"
1100 RETURN

```

PLAYER/MISSILE GRAPHICS
Using the Priority Register
JB 4/82

The priority of players and playfield objects can be controlled by setting bits in the priority register, PRIOR, location \$D01B. PRIOR has a RAM shadow, GPRIOR, at \$26F, or decimal 623. By poking different bits on at this location, you can control whether the player passes in front of or behind a playfield object of a particular color.

There are four types of priority, each of which is selected with one of the four least-significant bits of PRIOR. Bit D0 selects a mode in which all players pass in front of all playfield objects. Bit D1 selects a mode in which players 0 and 1 go in front, and players 2 and 4 go behind the playfield objects. When bit D2 is set, all playfield objects have priority over players, and when bit D3 is set, playfield objects 0 and 1 have priority over all players, which have priority over objects 2 and 3. In all cases, all players and all other playfield types have priority over the background and anything drawn in the background color. There is a chart of these priorities, along with some details on conflicting priorities, in Tech User Notes, C016555, on page III.8 of the Hardware Manual.

The following program shows priorities in action. A playfield is drawn, using all three colors, each color being a playfield object type. You select which player you want to use, 1-4. The program then asks you to select a priority. The choices, 1,2,4 or 8, are the numbers that can be poked into GPRIOR to turn on the appropriate bit. Once you have selected the priority, move the player across the different playfield objects, using the joystick. Move the joystick down to make the player appear the first time. When you press the trigger, you can select a new priority. To select a different player, press RESET and RUN the program again.

```

1 REM PRIORITY
2 REM JB 4/82
3 REM -- observe different priorities of players and playfield objects.
4 REM you select a player 1-4 and one of the four priority options.
5 REM then move the player over the different colors and see what happens.
6 REM *****
10 GOSUB 1000:REM initialize player
20 GOSUB 2000:REM draw playfield
30 UP=100:DOWN=200:EAST=300:WEST=400
40 SE=500:NE=550:SW=600:NW=650:REM motion routine locations
45 Z=90:REM no motion, return only
50 PRINT "PRIORITY 1,2,4,8...":INPUT P:POKE 623,P:REM select priority bit
60 ON STICK(0) GOSUB Z,Z,Z,Z,SE,NE,EAST,Z,SW,NW,WEST,Z,DOWN,UP,Z
70 IF STRIG(0)=0 THEN GOTO 50:REM on trigger, select new priority
80 GOTO 60:REM keep checking stick
90 RETURN:REM no motion, keep checking stick
95 REM *****
99 REM -- motion routines --
100 X=X-1:IF X<0 THEN X=0:REM *** move up
110 POKE PSTART+X,231:POKE PSTART+X+8,0:RETURN
111 REM draw top line, erase bottom line
200 X=X+1:IF X>250 THEN X=250:REM *** move down
210 POKE PSTART+X,231:POKE PSTART+X-8,0:RETURN
211 REM draw bottom line, erase top line
300 H=H+1:IF H>200 THEN H=200:REM *** move east
310 POKE HPOS,H:RETURN
400 H=H-1:IF H<50 THEN H=50:REM *** move west
410 POKE HPOS,H:RETURN

```



```

500 GOSUB DOWN:GOSUB EAST:RETURN:REM *** move southeast
550 GOSUB UP:GOSUB EAST:RETURN:REM *** move northeast
600 GOSUB DOWN:GOSUB WEST:RETURN:REM *** move southeast
650 GOSUB UP:GOSUB WEST:RETURN:REM *** move northwest
700 REM *****
999 REM this subroutine initializes the player
1000 GRAPHICS 5:PRINT "PLAYER 1,2,3,4...":INPUT PNUM:REM select a player
1005 REM assign start address, color register, horizontal position reg
1010 IF PNUM=1 THEN OFFSET=1024:COL=704:HPOS=53248
1020 IF PNUM=2 THEN OFFSET=1280:COL=705:HPOS=53249
1030 IF PNUM=3 THEN OFFSET=1536:COL=706:HPOS=53250
1040 IF PNUM=4 THEN OFFSET=1792:COL=707:HPOS=53251
1050 POKE 559,62:POKE 53277,3:REM enable players w/single-line resolution
1060 PMBASE=PEEK(106)-24:POKE 54279,PMBASE:REM step back 24 pages from
1061 REM ramtop to assign player ram area
1070 PSTART=PMBASE*256+OFFSET:REM starting address of player
1080 POKE COL,88:H=50:POKE HPOS,H:REM assign color, horizontal position
1090 FOR I=0 TO 255:POKE PSTART+I,0:NEXT I:REM clear player
1100 RETURN
1200 REM *****
1999 REM this subroutine draws the playfield-- one bar of each color
2000 COLOR 1
2010 FOR X=10 TO 20:FOR Y=0 TO 39
2020 PLOT X,Y
2030 NEXT Y:NEXT X
2040 COLOR 2
2050 FOR X=30 TO 40:FOR Y=0 TO 39
2060 PLOT X,Y
2070 NEXT Y:NEXT X
2080 COLOR 3
2090 FOR X=50 TO 60:FOR Y=0 TO 39
2100 PLOT X,Y
2110 NEXT Y:NEXT X
2120 RETURN

```

PLAYER/MISSILE GRAPHICS Collision Detection JB 5/82

When you are using Player/Missile Graphics, it is possible to detect collisions between players and missiles, players and other players, or between the playfield and either players or missiles. In order to do this, you must check the values at the special collision registers. The numbers reflect the bit patterns which tell you exactly which player, missile or playfield object has been hit.

There are 16 collision registers, and a special called HITCLR, which clears all of the other registers. HITCLR is write-only, which means you can only POKE it. If you check the PEEK, it does not match what you put there. POKEing anything other than a 0 into HITCLR (decimal location 53278) has the effect of clearing all collision registers.

The collision registers themselves are read-only. You cannot POKE into them. They are cleared by writing to HITCLR; this is the only way to change them. The contents of these registers reflect the state of the screen display. When any object occupies the same coordinates as any other object, the appropriate bit is turned on.

The 16 collision registers are located as follows:

```

53248 Missile 0 to Playfield
53249 Missile 1 to Playfield
53250 Missile 2 to Playfield
53251 Missile 3 to Playfield

53252 Player 0 to Playfield
53253 Player 1 to Playfield
53254 Player 2 to Playfield
53255 Player 3 to Playfield

53256 Missile 0 to Player
53257 Missile 1 to Player
53258 Missile 2 to Player
53259 Missile 3 to Player

53260 Player 0 to Player
53261 Player 1 to Player
53262 Player 2 to Player
53263 Player 3 to Player

```

The least significant nybble of each register is used to show collisions. The least significant bit, bit D0, is set (contains a 1) when there is a collision with Player or Playfield type 0. The next bit is set on a collision with Player

MILATARI * * * SEPTEMBER 1983

or Playfield type 1, and so on. For example, when Missile 1 collides with Player 3, location 53257 contains the binary number 0000 1000. The decimal equivalent is 8, so PEEK(53257)=8.

If several collisions happen before the registers are cleared, all of the affected bits stay on. The bit for a Player's collision with itself is always 0.

Playfield objects are objects drawn on the screen with regular Display List Graphics, as opposed to Player-Missile Graphics. Anything drawn with PLOT or DRAWTO is a playfield object.

The type of a Playfield object is determined by which color register it is drawn with. Objects drawn with register 0 are type 0, and collisions show up in bit D0. SETCOLOR numbers are the same as color register numbers.

In modes 2-7, color register 4 contains the background color. In modes 0 and 8 however, register 2 is the background. In these modes, a collision is always indicated between Playfield object 2 and all Players and Missiles that are on the screen.

```

1 REM COLLOSIONS
2 REM JB 5/82
3 REM -- detect collisions between a player and various playfield objects:
4 REM when a collision occurs, the playfield object changes color.
5 REM *****
10 GOSUB 1000:REM initialize player
20 GOSUB 2000:REM draw playfield
25 P0PF=53252:HITCLR=53278:REM location of collision & hitclear register
30 UP=100:DOWN=200:EAST=300:WEST=400
40 SE=500:NE=550:SW=600:NW=650:REM motion routine locations
45 Z=90:REM no motion, return only
50 GOSUB 3000:REM call collision checking routine
60 ON STICK(0) GOSUB Z,Z,Z,Z,SE,NE,EAST,Z,SW,NW,WEST,Z,DOWN,UP,Z
70 GOTO 50:REM keep checking
90 RETURN:REM no motion, keep checking stick
95 REM *****
99 REM -- motion routines --
100 X=X-1:IF X<0 THEN X=0:REM *** move up
110 POKE PSTART+X,231:POKE PSTART+X+8,0:RETURN
111 REM draw top line, erase bottom line
200 X=X+1:IF X>250 THEN X=250:REM *** move down
210 POKE PSTART+X,231:POKE PSTART+X-8,0:RETURN
211 REM draw bottom line, erase top line
300 H=H+1:IF H>200 THEN H=200:REM *** move east
310 POKE HPOS,H:RETURN
400 H=H-1:IF H<50 THEN H=50:REM *** move west
410 POKE HPOS,H:RETURN
500 GOSUB DOWN:GOSUB EAST:RETURN:REM *** move southeast
550 GOSUB UP:GOSUB EAST:RETURN:REM *** move northeast
600 GOSUB DOWN:GOSUB WEST:RETURN:REM *** move southwest
650 GOSUB UP:GOSUB WEST:RETURN:REM *** move northwest
700 REM *****
799 REM this subroutine initializes the player
1000 GRAPHICS 5+16:REM no text window
1005 REM assign start address, color register, horizontal position reg
1010 OFFSET=1024:COL=704:HPOS=53248:REM for player 0
1050 POKE 559,62:POKE 53277,3:REM enable players w/single-line resolution
1060 PMBASE=PEEK(106)-24:POKE 54279,PMBASE:REM step back 24 pages from
1061 REM ramtop to assign player ram area
1070 PSTART=PMBASE*256+OFFSET:REM starting address of player
1080 POKE COL,88:H=50:POKE HPOS,H:REM assign color, horizontal position
1090 FOR I=0 TO 255:POKE PSTART+I,0:NEXT I:REM clear player
1100 RETURN
1998 REM *****
1999 REM this subroutine draws the playfield-- one bar of each color
2000 COLOR 1
2010 FOR X=10 TO 20:FOR Y=0 TO 39
2020 PLOT X,Y
2030 NEXT Y:NEXT X
2040 COLOR 2
2050 FOR X=30 TO 40:FOR Y=20 TO 39
2060 PLOT X,Y
2070 NEXT Y:NEXT X
2080 COLOR 3
2090 FOR X=50 TO 60:FOR Y=10 TO 29
2100 PLOT X,Y
2110 NEXT Y:NEXT X
2120 RETURN
2998 REM *****
2999 REM check for collisions: if there is one, that object changes color

```


MILATARI * * * SEPTEMBER 1983

REVIEW OF THE ATARI PROGRAM-TEXT EDITOR

by Steve Booth

I highly recommend the ATARI Program-Text Editor for anyone who is attempting to perform sophisticated programming on the ATARI. There are some very powerful features built into this editor which make it an ideal choice. Features of this Editor include:

- 1) Full Support of the ATARI Character Set.
- 2) Ability to handle up to 200 characters per line. This is invaluable when trying to edit a LISTed Basic program which contains lines longer than 114 characters.
- 3) Single Keystroke commands to:
 - a) Move Cursor to Beginning of Line.
 - b) Move Cursor to End of Line.
 - c) Move Forward one Screen
 - d) Move Backward one Screen
- 4) Save Files in Standard DOS format.
- 5) Manipulation of a "Block of Lines".
 - a) Block Delete.
 - b) Block Move.
 - c) Block Copy.
 - d) Block Print - allows you to print your file without leaving the Editor.
 - e) Block Write - allows you to write some or all of your file to another disk file without leaving the Editor.
- 6) Ability to Merge other files into your current file.
- 7) Ability to perform String Searches (and change one string to another). There is also a Veto Option available.
- 8) Automatic Backup of a file is provided (so you never really destroy the original copy of the file).
- 9) If you accidentally Delete a Line, you can "recover" it.
- 10) The largest file that can be edited is about 325 sectors.
- 11) You do NOT have to press the RETURN key to make the change permanent.
- 12) The cost of this program is very reasonable (list is \$39.95).

HOWEVER, all is not "Sweetness and Light" when using this program. There are definite flaws in the program. These flaws include:

- 1) Poor Documentation. All the information is there; sometimes it takes a real miner to dig it out.
- 2) The concept of "buffers" -vs- "files" is not clearly stated. This could be frustrating to a new user.
- 3) The concept of the use of a command line is not clearly stated.
- 4) The fact that you cannot save the edit file to another disk should have been emphasized.
- 5) Commands that are typed on the command line could be more mnemonic (user-friendly). However, they ARE consistent in their usage.
- 5) Occasionally, the program will get "confused". This forces you to save what you have done and restart the program. This is a fairly trivial problem.

This is a short review of what this program can do. Overall, it is my belief that the capabilities of this program are outstanding, especially in view of the very modest price. I would suggest that you read the APX review to get a better feel for what this program can do for you.

— — — — —
(demopac 5 con't)

```
3000 C0=PEEK(708):C1=PEEK(709):C2=PEEK(710):REM get colors from registers
3010 IF PEEK(P0PF)=1 THEN C0=C0+1:IF C0>255 THEN C0=16:REM check object 0
3020 POKE 708,C0:REM put in new color (or same color if no collision)
3030 IF PEEK(P0PF)=2 THEN C1=C1+1:IF C1>255 THEN C1=16:REM check object 1
3040 POKE 709,C1:REM new color if collision
3050 IF PEEK(P0PF)=4 THEN C2=C2+1:IF C2>255 THEN C2=16:REM check object 2
3060 POKE 710,C2:REM new color if collision
3070 POKE HITCLR,1:REM clear collision register
3080 RETURN
```


DISK NEWS

by Steve Booth

This Month I have changed Disk 014 to include the Fix described in ANALOG 11 for FILL 'ER UP.

Hopefully, the Following will be New Disks in the Public Domain Library:

1) D025 - Games Disk

- STAR TREK - You are the Enterprise in this modified version of the HOUSTON ATARI User Group.
- INFO - The Documenting program that is on each new disk. This program will print out to the Screen or to the Printer, Documentation for each program on the disk!
- HOPPER - Similar to FROGGER only slower (it's written in BASIC).
- LIVE WIRE - this is the Version that can be used to make Binaries for either disk or cassette.
- MATH - a Math game for "the younger ones".
- LIGHT - a Light pen routine.
- STUNT MAN 2 - at last! a decent version of this program.

2) D026 - Utilities Disk

- INFO - See Above.
- AMODEM 3 - the Famous AMODEM program. One of the latest versions (if not the latest).
- AMDIAL - Automatic Dialing Program.
- AUTORUN - This program is automatically loaded in when the disk is booted, it contains the RS232 interface handler. This is needed for AMODEM3, AMDIAL, etc.
- DISKTRAN - a Disk Transfer Program (modem)
- CBOOTMGR - a program that does creates a disk file from a cassette?
- MICROPICS - Micro-Painter Pictures in BASIC (from ANALOG 12). The Loader program is also included.
- STRING - String Search Programs (from ANALOG 12) Both Assembler and Basic Versions are included.

and More ???

FOR SALE: Home Calc - powerful, inexpensive spreadsheet for home and business calculations.

Cassette - \$25.00

Diskette - \$35.00

Dave Frazer

542-7242

FINGER PRINT™

Gives you
fingertip
control
of your
Epson
Printer!

Review of 'FINGER PRINT'

by David Frazer

The 'finger print' you see to the left is not a clue for a new mystery adventure program. It is circuit board attachment for EPSON printers which allows you to select many of its features manually.

The product consists of a small circuit board with a 3 ROM chip mounted on it which needs to be installed inside your EPSON 80/100/F/T printer. An 8 page manual comes in the package with very easy to follow instructions and pictures to assist in the installation. Installing the board takes about 15 minutes.

Once the board is installed and tested, your next step is to learn its operation. Again, the well written manual makes this step an easily task. Using only the three control buttons on your EPSON (ON LINE, FF & LF), you can manually select up to 10 functions. The functions are 1) RESET - return to power-up defaults, 2) COMPRESSED - 17 cpi, 3) DOUBLE-WIDE, 4) EMPHASIZED, 5) DOUBLE-STRIKE, 6) PERFORATION SKIPOVER - creates top and bottom margin on each page, 7) INDENT - indents left margin 6 spaces, 8) 8 LINES PER INCH, 9) ITALICS, and 10) FINE PRINT - for printer with Grafrax-plus it set superscript mode and 12 characters per inch. Can be used in combination with compressed print for very fine print. Two additional features are BUFFER CLEAR which allows you to clear the printer buffer of unprinted information and ON-LINE PRESET which allows you to press FF for a form feed and have the printer automatically go on-line when the form feed is completed.

My favorites are the PERFORATION SKIPOVER (great for program listings) and ON-LINE PRESET. (No more waiting for a form feed to finish before setting the printer back on-line.)

'FINGER PRINT' is manufactured by Dresselhaus Computer Products, 837 E Alosta Ave, Glendora, CA 91740. (213)914-5831. Most area hardware dealers should have this product on the shelf. The list price is \$59.95.

=====

MILATARI Newsletter
David Frazer, Editor
P.O. Box 1191
Waukesha, WI 53187-1191

=====

Address Correction Requested

Forwarding and Return Postage Guaranteed

=====